

Ekstraoppgave 11.6.1.

a)

```
> with(plots)
```

Vi plotter først de to flatene $x^2 + y^2 = 1$ og $z = 4 - x$ for å få en ide om hvordan T ser ut.

```
> P1 := plot3d([x, sqrt(1 - x^2), z], x = 0..4, z = 0..4, color = blue, style = surface)
```

```
> P2 := plot3d([x, y, 4 - x], x = 0..4, y = 0..4, color = green, style = surface)
```

```
> display(P1, P2, axes = boxed, labels = [x, y, z])
```

Det var ikke særlig opplysende, men drei på figuren, så får du et bedre inntrykk.

Figuren må være avrenset av det grønne planet og den blå sylinderveggen.

I tillegg har T bunn nede i xy -planet, fra origo og ut til den blå sylinderveggen, og sidevegger i de to vertikale koordinatplanene, fra z -aksen og ut til den blå sylinderveggen.

Vi velger å integrere søylevis (altså i z -retning innerst). En søyle går fra xy -planet der $z = 0$ til det grønne planet der $z = 4 - x$.

For å få med alle søylene, må vi integrere over projeksjonen av T i xy -planet. Vi ser av figuren at denne projeksjonen er det samme som bunnen til T .

Vi kan faktisk se projeksjonen ved å vippe opp figuren slik at du har xy -planet snudd rett mot deg, mens z -aksen peker rett inn i skjermen.

Men det er jo også fort gjort å tegne den opp på en egen figur:

```
> implicitplot(x^2 + y^2 = 1, x = 0..1, y = 0..1)
```

Teller vi søylene kolonnevis, er det y som teller. I kolonnen ved x vil y gå fra $y = 0$ til $y = \sqrt{1 - x^2}$.

For å få med alle kolonnene må x gå fra $x = 0$ til $x = 1$.

Vi kan derfor skrive det søkte integralet som det itererte integralet
$$\int_{x=0}^1 \int_{y=0}^{\sqrt{1-x^2}} \int_{z=0}^{4-x} z \, dz \, dy \, dx.$$

Vi ber Maple beregne dette itererte integralet:

```
> int(int(int(z, z = 0..4 - x), y = 0..sqrt(1 - x^2)), x = 0..1)
```

En annen sak er at når vi har skrevet integralet som et iterert integral, kan vi skrive T som

$$T: \quad 0 \leq z \leq 4 - x, \quad 0 \leq y \leq \sqrt{1 - x^2}, \quad 0 \leq x \leq 1$$

og bruke det til å lage en mye bedre tegning av T . For at ikke de ulike sidene av T skal dekke over hverandre, gjør vi sidene transparente. Graden av transparens er et tall mellom 0 og 1, jo høyere tall, jo mer transparent.

```
> P1 := plot3d([x, sqrt(1 - x^2), z], x = 0..1, z = 0..4 - x, color = blue, style = surface, transparency = 0.5, numpoints = 5000)
> P2 := plot3d([x, y, 4 - x], x = 0..1, y = 0..sqrt(1 - x^2), color = green, style = surface, transparency = 0.5)
> P3 := plot3d([x, y, 0], x = 0..1, y = 0..sqrt(1 - x^2), color = yellow, style = surface, transparency = 0.5)
> P4 := plot3d([0, y, z], y = 0..1, z = 0..4, color = yellow, style = surface, transparency = 0.5)
> P5 := plot3d([x, 0, z], x = 0..1, z = 0..4 - x, color = yellow, style = surface, transparency = 0.5)
> display(P1, P2, P3, P4, P5, axes = framed, scaling = constrained, labels = [x, y, z])
```

Ved å snu og dreie på figuren, får du et godt inntrykk av hvordan den ser ut.

b)

Det ser ut til at flatene $y = x^2$ og $y = 6 + x$ er vertikale (z mangler i likningene), mens flatene $z = -3$ og $z = \sin x$ avgrenser T i z -retning. Vi tegner derfor først de to vertikale flatene:

```
> P1 := plot3d([x, 6 + x, z], x = -5..5, z = -3..5, style = surface, color = blue)
> P2 := plot3d([x, x^2, z], x = -5..5, z = -3..5, style = surface, color = green)
> display(P1, P2, axes = boxed)
```

Det er klart at både den grønne og den blå flaten er tegnet for langt, men vi lar det stå slik foreløpig. Så var det de to flatene som avgrenser T i z -retning:

```
> P3 := plot3d([x, y, -3], x = -5..5, y = -11..11, style = surface, color = red)
> P4 := plot3d([x, y, sin(x)], x = -5..5, y = -11..11, style = surface, color = yellow)
```

```
> display(P3, P4, axes = boxed)
```

Vi velger å integrere søylevis. Av den siste figuren ser vi at hver søyle starter på $z = -3$ og ender når den når den gule flaten, altså for $z = \sin x$.

For å få med alle søylene, kan vi se på projeksjonen av T i xy -planet, og den finner vi av den første figuren.

Projeksjonen er rett og slett området innenfor kurvene $y = x^2$ og $y = 6 + x$. Vi tegner den:

```
> implicitplot([y = x^2, y = 6 + x], x = -5 .. 5, y = -11 .. 11, color = [green, blue])
```

Vi velger å telle søylene kolonnevis. Kolonnen ved x går fra $y = x^2$ til $y = x + 6$.

For å få med alle kolonnene, trenger vi å vite x -koordinatene i de to skjæringspunktene på grafen over.

De lar vi Maple finne:

```
> solve(x^2 = x + 6, x)
```

Aha! For å få med alle kolonnene, må x gå fra $x = -2$ til $x = 3$.

Integralet er derved gitt ved følgende itererte integral:

$$\int_{x=-2}^3 \int_{y=x^2}^{x+6} \int_{z=-3}^{\sin x} x^2 \cdot y \cdot z \, dz \, dy \, dx.$$
 Vi lar Maple beregne det:

```
> int(int(int(x^2*y*z, z = -3 .. sin(x)), y = x^2 .. x + 6), x = -2 .. 3)
```

```
> simplify(%)
```

```
> evalf(%)
```

Når vi nå har skrevet integralet på iterert form, så vet vi også at T er gitt ved

$T: \quad -3 \leq z \leq \sin x, \quad x^2 \leq y \leq 6 + x, \quad -2 \leq x \leq 3,$

noe vi kan bruke til å tegne et bilde av T :

```
> P1 := plot3d([x, 6 + x, z], x = -2 .. 3, z = -3 .. sin(x), style = surface, color = blue, transparency = 0.5)
```

```
> P2 := plot3d([x, x^2, z], x = -2 .. 3, z = -3 .. sin(x), style = surface, color = green, transparency = 0.5)
```

```

> P3 := plot3d([x, y, -3], x=-2..3, y=x^2..x+6, style=surface, color=red, transparency=0.5)
> P4 := plot3d([x, y, sin(x)], x=-2..3, y=x^2..x+6, style=surface, color=yellow, transparency=0.5)
> display(P1, P2, P3, P4, axes=boxed)

```

Ekstraoppgave 11.6.2.

a)

```

> with(plots)

```

For å plote flatene, løser vi likningen for flaten med hensyn på en av de variable og skriver det slik:

```

> P1 := plot3d([x, y, x+y], x=-5..5, y=-5..5, color=blue)
> P2 := plot3d([x, y, 9-x^2], x=-5..5, y=-5..5, color=green)
> P3 := plot3d([x, 5, z], x=-5..5, z=-5..9, color=yellow)
> P4 := plot3d([x, -5, z], x=-5..5, z=-5..9, color=magenta)
> display(P1, P2, P3, P4, axes=boxed, labels=[x, y, z])

```

Noen av flatene ble for korte og noen for lange, men figuren gir likevel et visst inntrykk av området T .

Du kan dreie på figuren for å få et enda bedre inntrykk av hvordan flaten blir.

Vi velger å telle søylevis. Da starter z på den blå flaten $z = x + y$, og stopper når den når den grønne flaten $z = 9 - x^2$.

For å være sikker på at vi får med alle søylene én og bare én gang, trenger vi å se projeksjonen av T ned i xy -planet.

Denne projeksjonen ligger mellom de to vertikale veggene $y = 5$ og $y = -5$.

I x -retning er det skjæringskurvene mellom den grønne og den blå flaten som danner avgrensningen av projeksjonen.

Derfor eliminerer vi z fra likningene $z = x + y$ og $z = 9 - x^2$, og får $x + y = 9 - x^2$.

```

> P1 := implicitplot(x + y = 9 - x^2, x = -5 .. 5, y = -5 .. 5, color = blue)
> P2 := implicitplot(y = -5, x = -5 .. 5, y = -5 .. 5, color = magenta)
> P3 := implicitplot(y = 5, x = -5 .. 5, y = -5 .. 5, color = yellow)
> display(P1, P2, P3)

```

Vi velger å integrere linjevis.

Da vil x gå fra den blå kurven til venstre, helt til den når den blå kurven til høyre.

De to blå kurvene er gitt ved $x + y = 9 - x^2$. For å finne likningene for dem, løser vi denne likningen med hensyn på x :

```

> solve(x + y = 9 - x^2, x)

```

Med andre ord, x skal gå fra $-\frac{1}{2} - \frac{1}{2}\sqrt{37 - 4y}$ helt til den når $-\frac{1}{2} + \frac{1}{2}\sqrt{37 - 4y}$.

For å få med alle linjene med søyler, må y gå fra -5 til 5 .

Det itererte integralet vi skal beregne er altså
$$\int_{y=-5}^{y=5} \int_{x=-\frac{1}{2} - \frac{1}{2}\sqrt{37-4y}}^{-\frac{1}{2} + \frac{1}{2}\sqrt{37-4y}} \int_{z=x+y}^{z=9-x^2} f(x, y, z) \, dz \, dx \, dy.$$

```

> int(int(int(x^3, z = x + y .. 9 - x^2), x = -1/2 - 1/2*sqrt(37 - 4*y) .. -1/2 + 1/2*sqrt(37 - 4*y)), y = -5 .. 5)
> simplify(%)
> evalf(%)

```

Nå som vi har skrevet integralet som et iterert integral, kan vi også lage en penere illustrasjon av området T basert på beskrivelsen

$$T: \quad x + y \leq z \leq 9 - x^2, \quad -\frac{1}{2} - \frac{1}{2}\sqrt{37 - 4y} \leq x \leq -\frac{1}{2} + \frac{1}{2}\sqrt{37 - 4y}, \quad -5 \leq y \leq 5$$

```

> P1 := plot3d([x, y, x + y], x = -1/2 - 1/2*sqrt(37 - 4*y) .. -1/2 + 1/2*sqrt(37 - 4*y), y = -5 .. 5, color = blue)

```

```

| > P2 := plot3d([x, y, 9 - x^2], x = -1/2 - 1/2*sqrt(37 - 4*y) .. -1/2 + 1/2*sqrt(37 - 4*y), y = -5 .. 5, color = green)
|
| > P3 := plot3d([x, -5, z], x = -1/2 - 1/2*sqrt(37 - 4*(-5)) .. -1/2 + 1/2*sqrt(37 - 4*(-5)), z = x - 5 .. 9 - x^2, color = yellow)
|
| > P4 := plot3d([x, 5, z], x = -1/2 - 1/2*sqrt(37 - 4*5) .. -1/2 + 1/2*sqrt(37 - 4*5), z = x + 5 .. 9 - x^2, color = magenta)
|
| > display(P1, P2, P3, P4, axes = boxed, labels = [x, y, z])
|
| >

```